

## Intro 8 — Wald tests vary with nonlinear transforms

[Description](#)[Remarks and examples](#)[References](#)[Also see](#)

## Description

After fitting a DSGE model, we often perform tests of structural parameters, and these tests often place nonlinear restrictions on the parameters. The values and rejection rates of a Wald test for different nonlinear expressions of the same null hypothesis are different. We illustrate this issue, show that likelihood-ratio (LR) tests do not have this problem, and illustrate that you can parameterize your model in terms of invertible transforms of each parameter.

## Remarks and examples

[stata.com](#)

Remarks are presented under the following headings:

*Wald tests vary with nonlinear transforms*

*LR tests do not vary with nonlinear transforms*

## Wald tests vary with nonlinear transforms

Performing a statistical test of whether a structural parameter in a DSGE has a specific value is one of the most frequent forms of inference after `dsge` and `dsge1` estimation. The null hypothesis in one of these tests frequently places nonlinear restrictions on the underlying parameters. Two different nonlinear expressions of the same null hypothesis produce different Wald test statistics in finite samples and have different rejection rates. In other words, the Wald test is not invariant to nonlinear transforms of the null hypothesis. The LR test, on the other hand, is invariant to nonlinear transforms of the null hypothesis.

### ► Example 1: Different values from logically equivalent Wald tests

Equations (1)–(5) specify how the observed control variable inflation  $p_t$ , the unobserved control variable output growth  $y_t$ , and the observed control variable (interest rate)  $r_t$  depend on the states  $z_t$  and  $u_t$ , given the shocks  $\epsilon_t$  and  $\xi_t$ .

$$p_t = \beta E_t(p_{t+1}) + \kappa y_t \quad (1)$$

$$y_t = E_t(y_{t+1}) - \{r_t - E_t(p_{t+1}) - \rho z_t\} \quad (2)$$

$$r_t = (1/\beta)p_t + u_t \quad (3)$$

$$z_{t+1} = \rho z_t + \epsilon_{t+1} \quad (4)$$

$$u_{t+1} = \delta u_t + \xi_{t+1} \quad (5)$$

## 2 Intro 8 — Wald tests vary with nonlinear transforms

We estimate the parameters of this model using the macroeconomic data for the United States in `usmacro2.dta`.

```
. use https://www.stata-press.com/data/r18/usmacro2
(Federal Reserve Economic Data - St. Louis Fed, 2017-01-15)
. dsge (p = {beta}*F.p + {kappa}*y)
>      (y = F.y - (r - F.p - {rhoz}*z), unobserved)
>      (r = (1/{beta})*p + u)
>      (F.u = {rhoz}*u, state)
>      (F.z = {rhoz}*z, state)
(setting technique to bfgs)
Iteration 0: Log likelihood = -146218.64
Iteration 1: Log likelihood = -5532.4212 (backed up)
Iteration 2: Log likelihood = -1067.4665 (backed up)
Iteration 3: Log likelihood = -938.92415 (backed up)
Iteration 4: Log likelihood = -885.96401 (backed up)
(switching technique to nr)
Iteration 5: Log likelihood = -880.81743 (not concave)
Iteration 6: Log likelihood = -818.95369
Iteration 7: Log likelihood = -787.30327
Iteration 8: Log likelihood = -754.54306
Iteration 9: Log likelihood = -753.62794
Iteration 10: Log likelihood = -753.57273
Iteration 11: Log likelihood = -753.57131
Iteration 12: Log likelihood = -753.57131
DSGE model
Sample: 1955q1 thru 2015q4                                Number of obs = 244
Log likelihood = -753.57131
```

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
<hr/>						
<code>/structural</code>						
beta	.5146674	.0783489	6.57	0.000	.3611065	.6682283
kappa	.1659057	.0474074	3.50	0.000	.072989	.2588224
rhoz	.9545256	.0186424	51.20	0.000	.9179872	.991064
rhoz	.7005482	.0452604	15.48	0.000	.6118394	.789257
<hr/>						
sd(e.u)	2.318202	.3047435			1.720916	2.915489
sd(e.z)	.6507117	.1123844			.4304423	.8709811

The interest rate equation shown in (3) links the nominal interest rate to the inflation rate. The coefficient on inflation is  $1/\beta$ . We test whether this parameter is 1.5, a common benchmark value in the literature.

```
. testnl 1/_b[beta] = 1.5
(1) 1/_b[beta] = 1.5
          chi2(1) =          2.24
          Prob > chi2 =      0.1342
```

We do not reject the null hypothesis that  $1/\beta$  is 1.5.

If we test the logically equivalent hypothesis that  $\beta = 2/3$ , the statistic and  $p$ -value change.

```
. test _b[beta] =2/3
( 1)  [/structural]beta = .6666667
      chi2( 1) =      3.76
      Prob > chi2 =    0.0524
```

The values of these two logically equivalent Wald tests differ because Wald tests are not invariant to nonlinear transformation. This issue is well known in the literature; see [Gregory and Veall \(1985\)](#) and [Phillips and Park \(1988\)](#) for details. In this example, the inference of failing to reject the null hypothesis remains the same when using a 5% significance level, but this is not true in general. Different formulations of Wald tests can lead to different inferences.

◀

## LR tests do not vary with nonlinear transforms

### ▶ Example 2: LR tests are invariant to nonlinear transforms

We illustrate this feature by performing LR tests that  $\beta = 2/3$  and that  $1/\beta = 1.5$ . The current estimates are those of the unconstrained model. We repeat these results and store them as `unconstrained`.

```
. dsge
DSGE model
Sample: 1955q1 thru 2015q4
Log likelihood = -753.57131
Number of obs = 244
```

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
<b>/structural</b>						
beta	.5146674	.0783489	6.57	0.000	.3611065	.6682283
kappa	.1659057	.0474074	3.50	0.000	.072989	.2588224
rhoz	.9545256	.0186424	51.20	0.000	.9179872	.991064
rhou	.7005482	.0452604	15.48	0.000	.6118394	.789257
sd(e.u)	2.318202	.3047435			1.720916	2.915489
sd(e.z)	.6507117	.1123844			.4304423	.8709811

```
. estimates store unconstrained
```

#### 4 Intro 8 — Wald tests vary with nonlinear transforms

Now, we estimate the parameters of the constrained model in which  $\beta = 2/3$ , store the results as `constrained`, and perform an LR test of the null hypothesis that  $\beta = 2/3$ .

```
. constraint 1 _b[beta] = 2/3
. dsge (p = {beta}*F.p + {kappa}*y)
> (y = F.y - (r - F.p - {rhoz}*z), unobserved)
> (r = (1/{beta})*p + u)
> (F.u = {rhou}*u, state)
> (F.z = {rhoz}*z, state),
> constraint(1)
(setting technique to bfgs)
Iteration 0: Log likelihood = -119695.1
Iteration 1: Log likelihood = -1425.592 (backed up)
Iteration 2: Log likelihood = -984.57609 (backed up)
Iteration 3: Log likelihood = -948.41524 (backed up)
Iteration 4: Log likelihood = -945.83724 (backed up)
(switching technique to nr)
Iteration 5: Log likelihood = -945.06881 (backed up)
Iteration 6: Log likelihood = -760.71545
Iteration 7: Log likelihood = -755.52634
Iteration 8: Log likelihood = -755.11897
Iteration 9: Log likelihood = -755.11007
Iteration 10: Log likelihood = -755.11003
DSGE model
Sample: 1955q1 thru 2015q4                                Number of obs = 244
Log likelihood = -755.11003
( 1)  [/structural]beta = .6666667
```

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
/structural						
beta	.6666667	(constrained)				
kappa	.1076811	.0276892	3.89	0.000	.0534113	.1619509
rhoz	.9538522	.0187789	50.79	0.000	.9170462	.9906581
rhou	.7214328	.0439669	16.41	0.000	.6352593	.8076063
sd(e.u)	1.915459	.0867103			1.74551	2.085408
sd(e.z)	.4936797	.080513			.3358771	.6514822

```
. estimates store constrained
. lrtest unconstrained constrained
Likelihood-ratio test
Assumption: constrained nested within unconstrained
LR chi2(1) = 3.08
Prob > chi2 = 0.0794
```

Note that the value of the LR statistic is 3.08. We now illustrate an LR of the null hypothesis that  $1/\beta = 1.5$  produces the same value.

We cannot impose nonlinear restrictions on parameters, so we must begin by reparameterizing the unconstrained model by replacing  $\{\beta\}$  with  $1/\{\beta\}$ . To avoid having  $\{\beta\}$  mean two different things, we write the model in terms of  $\{\gamma\}=1/\{\beta\}$  and estimate the parameters:

```

. dsge (p = 1/{gamma}*F.p + {kappa}*y)
> (y = F.y - (r - F.p - {rhoz}*z), unobserved)
> (r = ({gamma})*p + u)
> (F.u = {rhoz}*u, state)
> (F.z = {rhoz}*z, state),
> from(gamma=2 kappa=0.15 rhoz=0.75 rhoz=0.95)
(setting technique to bfgs)
Iteration 0: Log likelihood = -1137.8808
Iteration 1: Log likelihood = -1097.9283 (backed up)
Iteration 2: Log likelihood = -1027.9554 (backed up)
Iteration 3: Log likelihood = -801.19555 (backed up)
Iteration 4: Log likelihood = -784.48041 (backed up)
(setting technique to nr)
Iteration 5: Log likelihood = -763.19407 (not concave)
Iteration 6: Log likelihood = -754.49971 (not concave)
Iteration 7: Log likelihood = -754.08362
Iteration 8: Log likelihood = -753.57362
Iteration 9: Log likelihood = -753.57131
Iteration 10: Log likelihood = -753.57131

```

DSGE model

Sample: 1955q1 thru 2015q4  
Log likelihood = -753.57131

Number of obs = 244

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
/structural						
gamma	1.943005	.2957867	6.57	0.000	1.363273	2.522736
kappa	.1659061	.0474073	3.50	0.000	.0729895	.2588226
rhoz	.9545256	.0186424	51.20	0.000	.9179872	.991064
rhoz	.7005481	.0452604	15.48	0.000	.6118393	.7892568
sd(e.u)	2.318205	.3047433			1.720919	2.915491
sd(e.z)	.6507124	.1123842			.4304434	.8709813

```
. estimates store unconstrained2
```

The estimates of the parameters other than `gamma` and the value of the log likelihood are nearly the same as those for the unconstrained model. The value for `gamma = 1.94` is the same as  $1/\beta = 1/0.514 = 1.95$ . By tightening the convergence tolerance, we could make these values exactly the same. These values are nearly the same because this example is an instance of a general property of maximum likelihood estimators. Transforming a parameter by an invertible function does not change the log likelihood or the other parameter estimates. In other words, maximum likelihood estimators are invariant to invertible transformations of the parameters; see [Casella and Berger \(2002, 319\)](#) for details.

Having stored the estimates from the unconstrained model, we now estimate the parameters of the constrained model and store these results in `constrained2`.

```
. constraint 2 _b[gamma] = 1.5
. dsge (p = 1/{gamma}*F.p + {kappa}*y)
> (y = F.y - (r - F.p - {rhoz}*z), unobserved)
> (r = ({gamma})*p + u)
> (F.u = {rhou}*u, state)
> (F.z = {rhoz}*z, state),
> constraint(2)
(setting technique to bfgs)
Iteration 0: Log likelihood = -119695.1
Iteration 1: Log likelihood = -1425.592 (backed up)
Iteration 2: Log likelihood = -984.57609 (backed up)
Iteration 3: Log likelihood = -948.41524 (backed up)
Iteration 4: Log likelihood = -945.83724 (backed up)
(switching technique to nr)
Iteration 5: Log likelihood = -945.06881 (backed up)
Iteration 6: Log likelihood = -760.71545
Iteration 7: Log likelihood = -755.52634
Iteration 8: Log likelihood = -755.11897
Iteration 9: Log likelihood = -755.11007
Iteration 10: Log likelihood = -755.11003
DSGE model
Sample: 1955q1 thru 2015q4                      Number of obs = 244
Log likelihood = -755.11003
( 1) [/structural]gamma = 1.5
```

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
/structural						
gamma	1.5	(constrained)				
kappa	.1076811	.0276892	3.89	0.000	.0534113	.1619509
rhoz	.9538522	.0187789	50.79	0.000	.9170462	.9906581
rhou	.7214328	.0439669	16.41	0.000	.6352593	.8076063
sd(e.u)	1.915459	.0867103			1.74551	2.085408
sd(e.z)	.4936797	.080513			.3358771	.6514822

```
. estimates store constrained2
```

The estimates of the parameters other than  $\gamma$  and the value of the log likelihood are the same as those for the constrained model. This is another instance of the invariance of the maximum likelihood estimator to invertible transformations of the parameters.

Having stored the log likelihoods from the constrained and unconstrained model, we now perform an LR of the null hypothesis that  $\gamma = 1.5$ .

```
. lrtest unconstrained2 constrained2
Likelihood-ratio test
Assumption: constrained2 nested within unconstrained2
LR chi2(1) = 3.08
Prob > chi2 = 0.0794
```

The LR test statistic and its  $p$ -value are the same as those reported for the test against the null hypothesis that  $\beta = 2/3$ , which illustrates that LR tests are invariant to nonlinear transforms.

## References

- Casella, G., and R. L. Berger. 2002. *Statistical Inference*. 2nd ed. Pacific Grove, CA: Duxbury.
- Gregory, A. W., and M. R. Veall. 1985. Formulating Wald tests of nonlinear restrictions. *Econometrica* 53: 1465–1468. <https://doi.org/10.2307/1913221>.
- Phillips, P. C. B., and J. Y. Park. 1988. On the formulation of Wald tests of nonlinear restrictions. *Econometrica* 56: 1065–1083. <https://doi.org/10.2307/1911359>.

## Also see

- [DSGE] [dsge postestimation](#) — Postestimation tools for dsge
- [DSGE] [dsgenl postestimation](#) — Postestimation tools for dsgenl
- [R] [lrtest](#) — Likelihood-ratio test after estimation
- [R] [test](#) — Test linear hypotheses after estimation

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).